// File: /contracts/MetadataTraits.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.7;

import "@openzeppelin/contracts/access/Ownable.sol";
import "./IMetadataTraits.sol";
import "./Structs.sol";

contract MetadataTraits is IMetadataTraits, Ownable {
    using Strings for uint8;
    using Strings for uint;

    string[] public _flockType = ['Free', 'Duet', 'Triplet', 'Quadruplet', 'Quintuplet', 'Sextuplet', 'Septuplet'
    , 'Free 2', 'Duet+', 'Triplet+', 'Quadruplet+', 'Quintuplet+', 'Sextuplet+', 'Septuplet+'
    , 'Mass', 'Opposing' ]; //todo not used
    string[] public _migrationType = ['Free', 'Southbound', 'Westbound', 'Northbound', 'Eastbound' ];
    string[] public _planeType = ['Jet', 'Single Prop', 'Twin Prop', 'Helicopter',
        'Hidden', 'Hidden', 'Hidden', 'Hidden', 'Hidden', 'Hidden'
    ];

    function setTypeNames(string[] memory typeNames) external virtual onlyOwner {
        require (_planeType.length == typeNames.length, "num strings incorrect");
        _planeType = typeNames;
    }

    function setTypeName(uint idx, string memory name) external virtual onlyOwner {
        require (idx < _planeType.length, "ioob");
        _planeType[idx] = name;
    }

    function getTraits(BaseAttributes memory atts, IArtData artData) external virtual override view returns (string memory)
    {

        return string.concat(
            '[{ "trait_type": "Sky Color", "value": "',
            artData.getSkyName(atts.skyCol),
            '" }, { "trait_type": "Proximity", "value": "',
            artData.getProximityName(atts.proximity),
            '" }, { "trait_type": "Num Planes", "value": "',
            atts.numPlanes.toString(),
            '" }, { "trait_type": "Lattice", "value": "',
            atts.extraParams[uint(EP.NumAngles)].toString(),
            '" }, { "trait_type": "Trails Palette", "value": "',
            artData.getColorPaletteName(atts.palette),
            '" }',

            getDynamicAttributesStr(atts, artData),

            ']');
    }

    struct CalcInformation {
        uint256 biggestFlockSize;
        uint8[] angleCounts;
        uint256 numTrailColors;
        uint256 numAngles;
        uint8[] planeTypeCounts;
        bytes directionBitCounts;
        bytes trailColorBitCounts;
        bytes speedBitCounts;
        uint256[] colorsByAngle;
        uint256[] speedsByAngle;
        uint8 mostCommonDirection;
        bool matchingSpeed;
    }

    function getDynamicAttributesStr(BaseAttributes memory art, IArtData artData)
        internal
        view
        virtual
        returns (string memory)
    {

        CalcInformation memory c;

        c.numAngles = art.extraParams[uint(EP.NumAngles)];
        c.angleCounts = new uint8[](c.numAngles);
        c.numTrailColors = artData.getPaletteSize(art.palette);
        c.planeTypeCounts = new uint8[](artData.getNumTypes());
        //array of single bytes. each byte being enough to store a bit for each of the 7 planes
        c.directionBitCounts = new bytes(c.numAngles);
        c.trailColorBitCounts = new bytes(c.numTrailColors);
        c.speedBitCounts = new bytes(artData.getNumSpeeds());
        c.colorsByAngle = new uint256[](c.numAngles);
        c.speedsByAngle = new uint256[](c.numAngles);

        for(uint8 i=0; i < art.numPlanes; i++) {
            PlaneAttributes memory planeAtts = art.planeAttributes[i];
            //keep count of how many planes you've seen with same angle
            c.angleCounts[planeAtts.angle] += 1;
            uint trailColIdx = planeAtts.trailCol % c.numTrailColors;
            //tally each plane type in this artwork
            c.planeTypeCounts[planeAtts.planeType] += 1;
            // implementation for figuring out if all planes sharing same direction also share same color
            c.trailColorBitCounts[trailColIdx] |= ( bytes1(0x01) << i);
            c.directionBitCounts[planeAtts.angle] |= ( bytes1(0x01) << i);
            c.speedBitCounts[planeAtts.speed] |= ( bytes1(0x01) << i);
            c.colorsByAngle[planeAtts.angle] = planeAtts.trailCol;
            c.speedsByAngle[planeAtts.angle] = planeAtts.speed;

        }

        c.biggestFlockSize = 1;
        for (uint8 i = 0; i < c.angleCounts.length; i++) {
            uint8 numInFlock = c.angleCounts[i];
            if ( numInFlock > c.biggestFlockSize) {
                c.mostCommonDirection = i;
                c.biggestFlockSize = numInFlock;
            }
        }

        c.matchingSpeed = isMatchingSpeed(c);

        string memory attsString;
        attsString = appendFlockTypeString(attsString, c, artData);
        attsString = appendMigrationTypeString(attsString, c, artData);
        attsString = appendMatchingTrailsStr(attsString, c);
        attsString = appendMatchingSpeedString(attsString, c);
        attsString = appendPlaneTypeCountsStr(attsString, c);

        return attsString;
    }

    function isMatchingSpeed(CalcInformation memory c) internal view virtual returns (bool) {
        uint256 speedForDirection = c.speedsByAngle[c.mostCommonDirection];
        return c.biggestFlockSize > 1 && (c.speedBitCounts[speedForDirection] == c.directionBitCounts[c.mostCommonDirection]);
    }

    function appendMatchingSpeedString(string memory attsString, CalcInformation memory c)
        internal view virtual returns (string memory)
    {
        return string.concat( attsString,
            composeAttributeString(
                'Matching Speed',
                c.matchingSpeed
                ? "Yes" : "No"
            ));
    }

    function appendFlockTypeString(string memory attsString, CalcInformation memory c, IArtData artData)
        internal view virtual returns (string memory)
    {
        //        uint8 flockTypeIdx = uint8((biggestFlockSize-1) + (art.matchingSpeed ? artData.getMaxNumPlanes() : 0));
        return string.concat( attsString,
            composeAttributeString('Flock Type',
            _flockType[uint8((c.biggestFlockSize-1) + (c.matchingSpeed ? artData.getMaxNumPlanes() : 0))]
            ));
    }

    function appendMigrationTypeString(string memory attsString, CalcInformation memory c, IArtData artData)
        internal view virtual returns (string memory)
    {
        // 4 possible directions, x angles.  so to get the general direction, divide by angles and multiply by 4
        // add 1 because first space is occupied by "free"
        uint8 migrationType = (uint8) (
            (c.biggestFlockSize == 1) ?
            0 : ((c.mostCommonDirection * 4) / c.numAngles ) + 1);

        return string.concat( attsString,
            composeAttributeString(
                'Migration Type',
                _migrationType[migrationType]
            ));

    }

    function appendMatchingTrailsStr(
        string memory attsString,
        CalcInformation memory c
    )
        internal
        pure
        virtual
        returns(string memory)
    {
        //this is matching trails for only the most common direction
        // by comparing the bit pattern (plane ids) for that direction with the bit pattern for a color seen for that direction
        uint256 colorForDirection = c.colorsByAngle[c.mostCommonDirection];

        return string.concat( attsString,
            composeAttributeString(
                'Matching Trails',
                c.biggestFlockSize > 1 && (c.trailColorBitCounts[colorForDirection] == c.directionBitCounts[c.mostCommonDirection])
                ? "Yes" : "No"
            ));
    }

    function appendPlaneTypeCountsStr(
        string memory attsString,
        CalcInformation memory c
    )
    internal
    view
    virtual
    returns(string memory)
    {
        for (uint8 i = 0; i < c.planeTypeCounts.length; i++) {
            if(c.planeTypeCounts[i] > 0) {
                attsString = string.concat( attsString,
                    composeAttributeString(
                        _planeType[i],
                        c.planeTypeCounts[i].toString()
                    ));
            }
        }

        return attsString;
    }

    function composeAttributeString(string memory trait, string memory value) internal pure returns (string memory) {
        return string.concat(
            ', { "trait_type": "', trait,
            '", "value": "', value,
            '" }'
        );

    }

}

// File: /contracts/Structs.sol



// File: contracts/Structs.sol

// SPDX-License-Identifier: MIT
pragma solidity ^0.8.12;

struct PlaneAttributes {
    uint8 locX;
    uint8 locY;
    uint8 angle;
    uint8 trailCol;
    uint8 level;
    uint8 speed;
    uint8 planeType;
    uint8[] extraParams;
}

struct BaseAttributes {
    uint8 proximity;
    uint8 skyCol;
    uint8 numPlanes;
    uint8 palette;
    PlaneAttributes[] planeAttributes;
    uint8[] extraParams;
}

enum EP {
    NumAngles
}


// File: /contracts/IMetadataTraits.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.7;

import "./Structs.sol";
import "./IArtData.sol";

interface IMetadataTraits {

    function getTraits(BaseAttributes memory atts, IArtData artData) external view returns (string memory);

}

// File: /contracts/IArtData.sol



// File: contracts/IArtData.sol


// SPDX-License-Identifier: MIT
pragma solidity >=0.4.22 <0.9.0;

import "@openzeppelin/contracts/utils/Strings.sol";
import "./IArtData.sol";

interface IArtData{

    struct ArtProps {
        uint256 numOfX;
        uint256 numOfY;
        uint256 numAngles;
        uint256 numTypes;
        uint256[] extraParams;
    }

    function getProps() external view returns(ArtProps memory);


    function getNumOfX() external view returns (uint) ;

    function getNumOfY() external view returns (uint);

    function getNumAngles() external view returns (uint);

    function getNumTypes() external view returns (uint);

    function getNumSpeeds() external view returns (uint);

    function getSkyName(uint index) external view returns (string calldata);

    function getNumSkyCols() external view returns (uint);

    function getColorPaletteName(uint paletteIdx) external view returns (string calldata) ;

    function getNumColorPalettes() external view returns (uint) ;

    function getPaletteSize(uint paletteIdx) external view returns (uint);

    function getProximityName(uint index) external view returns (string calldata);

    function getNumProximities() external view returns (uint);

    function getMaxNumPlanes() external view returns (uint);


    function getLevelRarities() external view returns (uint8[] calldata);

    function getSpeedRarities() external view returns (uint8[] calldata);

    function getPlaneTypeRarities() external view returns (uint8[] calldata);

    function getProximityRarities() external view returns (uint8[] calldata);

    function getSkyRarities() external view returns (uint8[] calldata) ;

    function getColorPaletteRarities() external view returns (uint8[] calldata) ;

}

// File: @openzeppelin/contracts/utils/Strings.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Strings.sol)

pragma solidity ^0.8.0;

/**
 * @dev String operations.
 */
library Strings {
    bytes16 private constant _HEX_SYMBOLS = "0123456789abcdef";

    /**
     * @dev Converts a `uint256` to its ASCII `string` decimal representation.
     */
    function toString(uint256 value) internal pure returns (string memory) {
        // Inspired by OraclizeAPI's implementation - MIT licence
        // https://github.com/oraclize/ethereum-api/blob/b42146b063c7d6ee1358846c198246239e9360e8/oraclizeAPI_0.4.25.sol

        if (value == 0) {
            return "0";
        }
        uint256 temp = value;
        uint256 digits;
        while (temp != 0) {
            digits++;
            temp /= 10;
        }
        bytes memory buffer = new bytes(digits);
        while (value != 0) {
            digits -= 1;
            buffer[digits] = bytes1(uint8(48 + uint256(value % 10)));
            value /= 10;
        }
        return string(buffer);
    }

    /**
     * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation.
     */
    function toHexString(uint256 value) internal pure returns (string memory) {
        if (value == 0) {
            return "0x00";
        }
        uint256 temp = value;
        uint256 length = 0;
        while (temp != 0) {
            length++;
            temp >>= 8;
        }
        return toHexString(value, length);
    }

    /**
     * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation with fixed length.
     */
    function toHexString(uint256 value, uint256 length) internal pure returns (string memory) {
        bytes memory buffer = new bytes(2 * length + 2);
        buffer[0] = "0";
        buffer[1] = "x";
        for (uint256 i = 2 * length + 1; i > 1; --i) {
            buffer[i] = _HEX_SYMBOLS[value & 0xf];
            value >>= 4;
        }
        require(value == 0, "Strings: hex length insufficient");
        return string(buffer);
    }
}

// File: @openzeppelin/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File: @openzeppelin/contracts/access/Ownable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (access/Ownable.sol)

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _transferOwnership(_msgSender());
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
        _;
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions anymore. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby removing any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}